
;; 1. Based on: run2
;; 2. Description: TMDD Prop error (A9) No PRD=999 No 200mg OMEGA BLOCK 3 SEX on V1
;; x1. Author: smartk1
;Project Name: CSF-1R

$PROB RUN#3 two comp iv infusion TMDD
$INPUT C ID TIME DOSE AMT DV LNDV MDV EVID PRD DAY CHT AGE SEX WT HT NOMT TAD RATE 
PAC CYCLE PVNS
$DATA nmpk04FEB14_v3.csv IGNORE=C IGNORE=(TIME.LT.0) IGNORE=(PRD.EQ.999) 
IGNORE=(DOSE.EQ.200)

$SUBROUTINES ADVAN13 TOL=10
$MODEL NCOMP=3
COMP=(CENTRAL,DEFDOSE,DEFOBS) 
COMP=(PERIPH)
COMP=(AUC)

$PK 
IF(SEX.EQ.0) THEN
GEND=1
ELSE
GEND=2
ENDIF

TVCL=THETA(1)
CL=TVCL*EXP(ETA(1))
TVV1=THETA(2)*(2-GEND)+THETA(7)*(GEND-1)
V1=TVV1*EXP(ETA(2))
TVVM=THETA(3)
VM=TVVM*EXP(ETA(3))
TVQ=THETA(4)
Q=TVQ*EXP(ETA(4))
TVV2=THETA(5)
V2=TVV2*EXP(ETA(5))
TVKM=THETA(6)
KM=TVKM*EXP(ETA(6))

S1=V1
S2=V2
K10=CL/V1
K12=Q/V1
K21=Q/V2

$DES 
DCP=A(1)/S1

DADT(1) = -A(1)*(K10+K12)+A(2)*K21-DCP*VM/(KM+DCP)
DADT(2) =  A(1)*K12-A(2)*K21
DADT(3) = DCP

$ERROR 
TY=A(1)/S1
AUC=A(3)

DEL = 0
IF (F.LE.0)   DEL=0.0001
IPRED=F+DEL

W1=THETA(8)



W=IPRED*W1
Y = IPRED + W*ERR(1)

IRES=IPRED-DV
IWRES = IRES/W

$THETA
(0,0.02)    ;[CL] ;L/h
(0,3)       ;[V1] ;L
(0,0.5)     ;[VM] ;(ug/mL)/h
(0,0.03)    ;[Q]  ;L/h
(0,3)       ;[V2] ;L
(0,0.5)     ;[KM] ;ug/mL
(0,1)       ;[V1SEX1]
(0.1)       ;[W1]

$OMEGA 
0.2 ;[BSV CL]

$OMEGA BLOCK (3)
0.05 ;[BSV V1]
0.07 0.4 ;[BSV VM]
0.05 0.1 0.3 ;[BSV Q]

$OMEGA
0.3 ;[BSV V2]
0.8 ;[BSV KM]

$SIGMA 
1 FIX ;[P] PROPORTIONAL COMPONENT

$EST MAXEVAL=9999 MSFO=run3.msf PRINT=5 METHOD=1 INTER POSTHOC NOABORT SIG=3 
NOOMEGABOUNDTEST NOTHETABOUNDTEST
$COVARIANCE MAT=S PRINT=E UNCONDITIONAL
$TABLE DOSE AMT ID TIME TAD MDV TY IPRED IWRES IRES CWRES AUC ONEHEADER NOPRINT 
FILE=sdtab3
$TABLE ID CL V2 Q V1 VM KM ETA1 ETA2 ETA3 ETA4 ETA5 ETA6 ONEHEADER NOAPPEND NOPRINT
FILE=patab3
$TABLE DOSE ID SEX CHT PRD PAC CYCLE PVNS ONEHEADER NOAPPEND NOPRINT FILE=catab3
$TABLE ID HT WT AGE ONEHEADER NOAPPEND NOPRINT FILE=cotab3

; Problem no. 2
$PROBLEM    Model simulations
$INPUT C ID TIME DOSE AMT DV LNDV MDV EVID PRD DAY CHT AGE SEX WT HT NOMT TAD RATE 
PAC CYCLE PVNS
$DATA nmpk04FEB14_v3.csv IGNORE=C IGNORE=(TIME.LT.0) IGNORE=(PRD.EQ.999) 
IGNORE=(DOSE.EQ.200) REWIND
$MSFI run3.msf
$SIMULATION (20200520) ONLYSIM NSUB=200 TRUE=FINAL
$TABLE DOSE ID SEX CHT PRD PAC CYCLE HT WT AGE TAD TIME EVID MDV TY DV IPRED 
ONEHEADER NOPRINT FILE=simtab3


